pragma solidity 0.4.24;


/**
 * @title SafeMath
 * @dev Math operations with safety checks that throw on error
 */
library SafeMath {

  /**
  * @dev Multiplies two numbers, throws on overflow.
  */
  function mul(uint256 _a, uint256 _b) internal pure returns (uint256 c) {
    // Gas optimization: this is cheaper than asserting 'a' not being zero, but the
    // benefit is lost if 'b' is also tested.
    // See: https://github.com/OpenZeppelin/openzeppelin-solidity/pull/522
    if (_a == 0) {
      return 0;
    }

    c = _a * _b;
    assert(c / _a == _b);
    return c;
  }

  /**
  * @dev Integer division of two numbers, truncating the quotient.
  */
  function div(uint256 _a, uint256 _b) internal pure returns (uint256) {
    // assert(_b > 0); // Solidity automatically throws when dividing by 0
    // uint256 c = _a / _b;
    // assert(_a == _b * c + _a % _b); // There is no case in which this doesn't hold
    return _a / _b;
  }

  /**
  * @dev Subtracts two numbers, throws on overflow (i.e. if subtrahend is greater than minuend).
  */
  function sub(uint256 _a, uint256 _b) internal pure returns (uint256) {
    assert(_b <= _a);
    return _a - _b;
  }

  /**
  * @dev Adds two numbers, throws on overflow.
  */
  function add(uint256 _a, uint256 _b) internal pure returns (uint256 c) {
    c = _a + _b;
    assert(c >= _a);
    return c;
  }
}

/**
 * @title ERC20Basic
 * @dev Simpler version of ERC20 interface
 * See https://github.com/ethereum/EIPs/issues/179
 */
contract ERC20Basic {
  function totalSupply() public view returns (uint256);
  function balanceOf(address _who) public view returns (uint256);
  function transfer(address _to, uint256 _value) public returns (bool);
  event Transfer(address indexed from, address indexed to, uint256 value);
}

/**
 * @title Basic token
 * @dev Basic version of StandardToken, with no allowances.
 */
contract BasicToken is ERC20Basic {
  using SafeMath for uint256;

  mapping(address => uint256) internal balances;

  uint256 internal totalSupply_;

  /**
  * @dev Total number of tokens in existence
  */
  function totalSupply() public view returns (uint256) {
    return totalSupply_;
  }

  /**
  * @dev Transfer token for a specified address
  * @param _to The address to transfer to.
  * @param _value The amount to be transferred.
  */
  function transfer(address _to, uint256 _value) public returns (bool) {
    require(_value <= balances[msg.sender]);
    require(_to != address(0));

    balances[msg.sender] = balances[msg.sender].sub(_value);
    balances[_to] = balances[_to].add(_value);
    emit Transfer(msg.sender, _to, _value);
    return true;
  }

  /**
  * @dev Gets the balance of the specified address.
  * @param _owner The address to query the the balance of.
  * @return An uint256 representing the amount owned by the passed address.
  */
  function balanceOf(address _owner) public view returns (uint256) {
    return balances[_owner];
  }

}


/**
 * @title Burnable Token
 * @dev Token that can be irreversibly burned (destroyed).
 */
contract BurnableToken is BasicToken {

  event Burn(address indexed burner, uint256 value);

  /**
   * @dev Burns a specific amount of tokens.
   * @param _value The amount of token to be burned.
   */
  function burn(uint256 _value) public {
    _burn(msg.sender, _value);
  }

  function _burn(address _who, uint256 _value) internal {
    require(_value <= balances[_who]);
    // no need to require value <= totalSupply, since that would imply the
    // sender's balance is greater than the totalSupply, which *should* be an assertion failure

    balances[_who] = balances[_who].sub(_value);
    totalSupply_ = totalSupply_.sub(_value);
    emit Burn(_who, _value);
    emit Transfer(_who, address(0), _value);
  }
}

/**
 * @title ERC20 interface
 * @dev see https://github.com/ethereum/EIPs/issues/20
 */
contract ERC20 is ERC20Basic {
  function allowance(address _owner, address _spender)
    public view returns (uint256);

  function transferFrom(address _from, address _to, uint256 _value)
    public returns (bool);

  function approve(address _spender, uint256 _value) public returns (bool);
  event Approval(
    address indexed owner,
    address indexed spender,
    uint256 value
  );
}

/**
 * @title Standard ERC20 token
 *
 * @dev Implementation of the basic standard token.
 * https://github.com/ethereum/EIPs/issues/20
 * Based on code by FirstBlood: https://github.com/Firstbloodio/token/blob/master/smart_contract/FirstBloodToken.sol
 */
contract StandardToken is ERC20, BasicToken {

  mapping (address => mapping (address => uint256)) internal allowed;


  /**
   * @dev Transfer tokens from one address to another
   * @param _from address The address which you want to send tokens from
   * @param _to address The address which you want to transfer to
   * @param _value uint256 the amount of tokens to be transferred
   */
  function transferFrom(
    address _from,
    address _to,
    uint256 _value
  )
    public
    returns (bool)
  {
    require(_value <= balances[_from]);
    require(_value <= allowed[_from][msg.sender]);
    require(_to != address(0));

    balances[_from] = balances[_from].sub(_value);
    balances[_to] = balances[_to].add(_value);
    allowed[_from][msg.sender] = allowed[_from][msg.sender].sub(_value);
    emit Transfer(_from, _to, _value);
    return true;
  }

  /**
   * @dev Approve the passed address to spend the specified amount of tokens on behalf of msg.sender.
   * Beware that changing an allowance with this method brings the risk that someone may use both the old
   * and the new allowance by unfortunate transaction ordering. One possible solution to mitigate this
   * race condition is to first reduce the spender's allowance to 0 and set the desired value afterwards:
   * https://github.com/ethereum/EIPs/issues/20#issuecomment-263524729
   * @param _spender The address which will spend the funds.
   * @param _value The amount of tokens to be spent.
   */
  function approve(address _spender, uint256 _value) public returns (bool) {
    allowed[msg.sender][_spender] = _value;
    emit Approval(msg.sender, _spender, _value);
    return true;
  }

  /**
   * @dev Function to check the amount of tokens that an owner allowed to a spender.
   * @param _owner address The address which owns the funds.
   * @param _spender address The address which will spend the funds.
   * @return A uint256 specifying the amount of tokens still available for the spender.
   */
  function allowance(
    address _owner,
    address _spender
   )
    public
    view
    returns (uint256)
  {
    return allowed[_owner][_spender];
  }

  /**
   * @dev Increase the amount of tokens that an owner allowed to a spender.
   * approve should be called when allowed[_spender] == 0. To increment
   * allowed value is better to use this function to avoid 2 calls (and wait until
   * the first transaction is mined)
   * From MonolithDAO Token.sol
   * @param _spender The address which will spend the funds.
   * @param _addedValue The amount of tokens to increase the allowance by.
   */
  function increaseApproval(
    address _spender,
    uint256 _addedValue
  )
    public
    returns (bool)
  {
    allowed[msg.sender][_spender] = (
      allowed[msg.sender][_spender].add(_addedValue));
    emit Approval(msg.sender, _spender, allowed[msg.sender][_spender]);
    return true;
  }

  /**
   * @dev Decrease the amount of tokens that an owner allowed to a spender.
   * approve should be called when allowed[_spender] == 0. To decrement
   * allowed value is better to use this function to avoid 2 calls (and wait until
   * the first transaction is mined)
   * From MonolithDAO Token.sol
   * @param _spender The address which will spend the funds.
   * @param _subtractedValue The amount of tokens to decrease the allowance by.
   */
  function decreaseApproval(
    address _spender,
    uint256 _subtractedValue
  )
    public
    returns (bool)
  {
    uint256 oldValue = allowed[msg.sender][_spender];
    if (_subtractedValue >= oldValue) {
      allowed[msg.sender][_spender] = 0;
    } else {
      allowed[msg.sender][_spender] = oldValue.sub(_subtractedValue);
    }
    emit Approval(msg.sender, _spender, allowed[msg.sender][_spender]);
    return true;
  }

}


/**
 * @title Ownable
 * @dev The Ownable contract has an owner address, and provides basic authorization control
 * functions, this simplifies the implementation of "user permissions".
 */
contract Ownable {
  address public owner;


  event OwnershipRenounced(address indexed previousOwner);
  event OwnershipTransferred(
    address indexed previousOwner,
    address indexed newOwner
  );


  /**
   * @dev The Ownable constructor sets the original `owner` of the contract to the sender
   * account.
   */
  constructor() public {
    owner = msg.sender;
  }

  /**
   * @dev Throws if called by any account other than the owner.
   */
  modifier onlyOwner() {
    require(msg.sender == owner);
    _;
  }

  /**
   * @dev Allows the current owner to relinquish control of the contract.
   * @notice Renouncing to ownership will leave the contract without an owner.
   * It will not be possible to call the functions with the `onlyOwner`
   * modifier anymore.
   */
  function renounceOwnership() public onlyOwner {
    emit OwnershipRenounced(owner);
    owner = address(0);
  }

  /**
   * @dev Allows the current owner to transfer control of the contract to a newOwner.
   * @param _newOwner The address to transfer ownership to.
   */
  function transferOwnership(address _newOwner) public onlyOwner {
    _transferOwnership(_newOwner);
  }

  /**
   * @dev Transfers control of the contract to a newOwner.
   * @param _newOwner The address to transfer ownership to.
   */
  function _transferOwnership(address _newOwner) internal {
    require(_newOwner != address(0));
    emit OwnershipTransferred(owner, _newOwner);
    owner = _newOwner;
  }
}

/**
* @title CTF15Token
* @dev Very simple ERC20 Token example, where all tokens are pre-assigned to the creator.
* Note they can later distribute these tokens as they wish using `transfer` and other
* `StandardToken` functions.
*/
contract CTF15Token is StandardToken, BurnableToken, Ownable {
        address public owner;
        address public pricingBot;
        string public constant name = "CTF15Token"; // solium-disable-line uppercase
        string public constant symbol = "C15"; // solium-disable-line uppercase
        uint8 public constant decimals = 5; // solium-disable-line uppercase

        uint256 public constant INITIAL_SUPPLY = 0 * (10 ** uint256(decimals));

        /**
        * @dev Constructor that gives msg.sender all of existing tokens.
        */
        constructor() public {
                owner = msg.sender;
                pricingBot = msg.sender;
                coinNumber = 15;
                // Prices in XXX - USD
                coinPrices = new uint64[](coinNumber);
                // Shares go from 1 to 1000, i.e. pro mille
                coinShares = new uint64[](coinNumber);
        }

        /**
        * CTF15 coin prices
        */
        uint8 public coinNumber;
        uint64 public buyBackPrice;
        uint64[] public coinPrices;
        uint64[] public coinShares;
        uint64 public indexValue;

        /*
        * True if Minting period finished
        */
        bool public mintingFinished = false;

        /**
        * CTF15 specific Events
        */
        event BuyAssets(
                uint256[] assetsToBuy
        );
        event SellAssets(
                uint256[] assetsToSell
        );
        event Mint(
                address indexed to,
                uint256 amount
        );
        event MintFinished();

        modifier canMint() {
                require(!mintingFinished);
                _;
        }

        modifier hasMintPermission() {
                require(msg.sender == owner);
                _;
        }

        /**
        * @dev Function to mint tokens
        * @param _to The address that will receive the minted tokens.
        * @param _amount The amount of tokens to mint.
        * @return A boolean that indicates if the operation was successful.
        */
        function mint(
                address _to,
                uint256 _amount
        )
        hasMintPermission
        canMint
        public
        returns (bool) {
                totalSupply_ = totalSupply_.add(_amount);
                balances[_to] = balances[_to].add(_amount);
                emit Mint(_to, _amount);
                emit Transfer(address(0), _to, _amount);
                return true;
        }

        /**
        * @dev Function to stop minting new tokens.
        * @return True if the operation was successful.
        */
        function finishMinting() onlyOwner canMint public returns (bool) {
                mintingFinished = true;
                emit MintFinished();
                return true;
        }

        /**
        * Functions for CTF15
        */
        function UpdatePricingBot(address newBotAddress) onlyOwner public {
                pricingBot = newBotAddress;
        }

        function UpdateCoinShares(uint64[] _shares) public {
                require(msg.sender == pricingBot);
                require(_shares.length >= coinNumber);

                for (uint8 i = 0; i < coinNumber; i++) {
                        coinShares[i] = _shares[i];
                }
        }

        function UpdateCoinPrices(uint64[] _prices) public {
                require(msg.sender == pricingBot);
                require(_prices.length >= coinNumber);

                for (uint8 i = 0; i < coinNumber; i++) {
                        coinPrices[i] = _prices[i];
                }
        }

        // ethIndex is the index (0-14) of Ethereum in our basket
        function GenerateNewTokens(uint ethIndex, uint totalAmount) public returns(bool) {
                // Set new owner values of token and mint them
                mint(msg.sender, totalAmount);

                // Generate data for trading bot which assets to buy
                uint256[] memory buyOrders = new uint256[](coinNumber);
                for (uint8 i = 0; i < coinNumber; i++) {
                        buyOrders[i] = (coinShares[i] * totalAmount * coinPrices[ethIndex]) / (1e16 * 1000);
                }
                emit BuyAssets(buyOrders);
                return true;
        }

        function SellTokens(uint ethIndex, uint amount) public returns(bool) {
                // Ensure sender has enough funds to Sell shares
                require(amount <= balances[msg.sender]);

                // Set new balance
                _burn(msg.sender, amount);

                // Generate data for trading bot which assets to sell
                uint256[] memory sellOrders = new uint256[](coinNumber);
                for (uint8 i = 0; i < coinNumber; i++) {
                        sellOrders[i] = (coinShares[i] * amount * coinPrices[ethIndex]) / (1e16 * 1000);
                }
                emit SellAssets(sellOrders);
                return true;
        }

        function SetBuyBackPrice(uint64 newPrice) public onlyOwner {
                buyBackPrice = newPrice;
        }

        function SetCoinNumber(uint8 newNumber) public onlyOwner {
                coinNumber = newNumber;
                coinPrices = new uint64[](coinNumber);
                coinShares = new uint64[](coinNumber);
        }

        function SetIndexValue(uint64 newValue) public onlyOwner {
                indexValue = newValue;
        }
}